



Министерство науки и высшего образования Российской Федерации  
Федеральное государственное бюджетное образовательное учреждение  
высшего образования  
«Московский государственный технический университет  
имени Н.Э. Баумана  
(национальный исследовательский университет)»  
(МГТУ им. Н.Э. Баумана)

---

ФАКУЛЬТЕТ «Информатика и системы управления»

КАФЕДРА «Программное обеспечение ЭВМ и информационные технологии»

---

# Отчёт

## по лабораторной работе № 7

**Название:** Определение вероятности отказа при помощи GPSS

**Дисциплина:** Моделирование

Студент

ИУ7-75Б

(Группа)

Д.В. Сусликов

(Подпись, дата)

(И.О. Фамилия)

Преподаватель

И.В. Рудаков

(Подпись, дата)

(И.О. Фамилия)


*Москва, 2021*

## **Содержание**

<b>Задание</b>	<b>3</b>
<b>Аналитическая часть</b>	<b>4</b>
<b>Результаты работы</b>	<b>5</b>
<b>Вывод</b>	<b>5</b>
<b>Листинги</b>	<b>6</b>

## Задание

В информационный центр приходят клиенты через интервал времени  $10 \pm 2$  минуты. Если все три имеющихся оператора заняты, клиенту отказывают в обслуживании. Операторы имеют разную производительность и могут обеспечивать обслуживание среднего запроса пользователя за  $20 \pm 5$ ;  $40 \pm 10$ ;  $40 \pm 20$ . Клиенты стремятся занять свободного оператора с максимальной производительностью. Полученные запросы сдаются в накопитель. Откуда выбираются на обработку. На первый компьютер запросы от 1 и 2-ого операторов, на второй – запросы от 3-его. Время обработки запросов первым и 2-м компьютером равны соответственно 15 и 30 мин. Промоделировать процесс обработки 300 запросов. Для выполнения поставленного задания необходимо создать концептуальную модель в терминах СМО, определить эндогенные и экзогенные переменные и уравнения модели. За единицу системного времени выбрать 0,01 минуты.




## Аналитическая часть

В процессе взаимодействия клиентов с информационным центром возможно:

- 1) Режим нормального обслуживания, т.е. клиент выбирает одного из свободных операторов, отдавая предпочтение тому у которого меньше номер.
- 2) Режим отказа в обслуживании клиента, когда все операторы заняты.

### Переменные и уравнения имитационной модели

Эндогенные переменные: время обработки задания  $i$ -ым оператором, время решения этого задания  $j$ -ым компьютером. Экзогенные переменные: число обслуженных клиентов и число клиентов, получивших отказ.



$$P_{\text{отк}} = \frac{C_{\text{отк}}}{C_{\text{отк}} + C_{\text{обсл}}}$$

## Результаты работы

Ниже представлены результаты работы программы.

FACILITY	ENTRIES	UTIL.	AVE. TIME	AVAIL.	OWNER	PEND	INTER	RETRY	DELAY
OPER1	121	0.788	19.924	1	0	0	0	0	0
OPER2	59	0.772	40.036	1	0	0	0	0	0
OPER3	51	0.711	42.640	1	0	0	0	0	0
SPC1	180	0.883	15.000	1	0	0	0	0	0
SPC2	51	0.500	30.000	1	0	0	0	0	0

QUEUE	MAX	CONT.	ENTRY	ENTRY(0)	AVE.CONT.	AVE.TIME	AVE.(-0)	RETRY
PC1_QUEUE	2	0	180	61	0.279	4.737	7.165	0
PC2_QUEUE	1	0	51	48	0.004	0.212	3.598	0

SAVEVALUE	RETRY	VALUE
TRANS_PROCESSED	0	231.000
TRANS_DROPPED	0	69.000
TRANS_DROPPED_PROB	0	0.230

**Рисунок 1 – Результаты работы**

## Вывод

Процент потерь равен 23%.

## Листинги

```
SIMULATE
GENERATE 10,2,,300,      ; блок GENERATE осуществляет ввод транзактов в модель
; A      средний интервал времени между последовательными поступлениями транзактов в модель
; [B]     модификатор, который изменяет значения интервала генерации транзактов
; по сравнению с интервалом, указанным операндом A
; [C]     задержка в выработке первого транзакта (0)
; [D]     число вырабатываемых источником заявок
; [E]     приоритет заявок

; если первый оператор занят, переход ко второму
OP1      GATE NU  OPER1,OP2
; блок GATE определяет состояние устройства
; оператор задает условие пропуска транзакта
;          NU          устройство не используется (NOT USED)
; A      операнд задает устройство для проверки
; [B]     операнд задает блок, в который перейдет транзакт, если оператор вернет "FALSE"

        SEIZE     OPER1      ; транзакт занимает устройство
        ADVANCE   20,5        ; задержка транзакта в течение некоторого времени
        RELEASE   OPER1      ; освобождение устройства
        TRANSFER  ,PC1,,      ; переход в блок первого компьютера
; A задает режим выполнения блока (условие)
; B и C - метки блоков
; D - индексная константа N(проверка входа в блоки B + xN)

; если второй оператор занят, переход к третьему
OP2      GATE NU  OPER2,OP3
        SEIZE     OPER2      ; транзакт занимает устройство
        ADVANCE   40,10      ; задержка транзакта
        RELEASE   OPER2      ; устройство освобождается
        TRANSFER  ,PC1        ; переход в блок первого компьютера

; если и третий оператор занят, заявка не обслуживается
OP3      GATE NU  OPER3,DROP
        SEIZE     OPER3
        ADVANCE   40,20      ; задержка транзакта
        RELEASE   OPER3
        TRANSFER  ,PC2        ; переход в блок второго компьютера
```

```

PC1      QUEUE      PC1_QUEUE      ; постановка транзакта в очередь
        SEIZE       SPC1            ; транзакт занимает устройство
        DEPART      PC1_QUEUE      ; извлечение транзакта из очереди
        ADVANCE      15             ; задержка транзакта
        RELEASE      SPC1           ; освобождение устройства
        TRANSFER     ,PROC          ; транзакт обслужен, переход к завершению

PC2      QUEUE      PC2_QUEUE      ; постановка транзакта в очередь
        SEIZE       SPC2            ; транзакт занимает устройство
        DEPART      PC2_QUEUE      ; извлечение транзакта из очереди
        ADVANCE      30             ; задержка транзакта
        RELEASE      SPC2           ; освобождение устройства
        TRANSFER     ,PROC          ; транзакт обслужен, переход к завершению

PROC     TRANSFER    ,ENDING
DROP     TRANSFER    ,ENDING

        ; количество обработанных заявок
ENDING   SAVEVALUE   TRANS_PROCESSED,N$PROC
        SAVEVALUE   TRANS_DROPPED,N$DROP
        ; вероятность потери заявки
        SAVEVALUE   TRANS_DROPPED_PROB, ((N$DROP) / (N$ENDING))

        TERMINATE 1
        START 300

```